```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

We consider a time-delay model of a flexible rod with a torque studied in H. Mounier, J. Rudolph, M. Petitot, M. Fliess, "A flexible rod as a linear delay system", in *Proccedings of 3rd European Control Conference*, Rome (Italy), 1995, and defined by the following system matrix:

```
> A:=DefineOreAlgebra(diff=[d,t],dual_shift=[delta,s],polynom=[t,s]):
> R:=matrix(2,3,[d,-d*delta,-1,2*d*delta,-d*delta^2-d,0]);
```

$$R := \begin{bmatrix} d & -d\,\delta & -1 \\ 2\,d\,\delta & -d\,\delta^2 - d & 0 \end{bmatrix}$$

If we denote by $A = \mathbb{Q}[d, \delta]$ the commutative polynomial ring of differential time-delay operators with rational constant coefficients, $M = A^{1\times 3}/(A^{1\times 2}\,R)$ the A-module finitely presented by R, then we can compute the A-module structure of the endomorphism ring $E = \mathrm{end}_A(M)$ of M:

```
> Endo:=MorphismsConstCoeff(R,R,A,mult_table):
> Endo[1];
```

$$[\begin{bmatrix} d & 0 & 0 \\ 0 & d & 0 \\ 0 & 0 & d \end{bmatrix}, \begin{bmatrix} 0 & 0 & -1-\delta^2 \\ 0 & 0 & -2\,\delta \\ 0 & 0 & d\,\delta^2 - d \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ -2\,\delta & 1+\delta^2 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & d & -\delta \\ d & 0 & -1 \\ 0 & 0 & d\,\delta \end{bmatrix},$$
$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 1+\delta^2 & 0 \\ 0 & 2\,\delta & 0 \\ 0 & 0 & 2\,\delta \end{bmatrix}, \begin{bmatrix} 0 & -d & \delta \\ -d & 0 & 1 \\ 0 & 0 & -d\,\delta \end{bmatrix}]$$

```
> Endo[2];
```

$$\begin{bmatrix} 1 & 1 & 0 & 0 & 0 & 0 & \delta \\ \delta & 0 & 0 & 0 & 0 & 0 & 1 \\ -1 & 0 & 0 & 0 & d & 0 & 0 \\ 0 & 0 & d & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & d & 2 \end{bmatrix}$$

```
> Endo[3];
```

$$\begin{bmatrix}
[1,1] & d & 0 & 0 & 0 & 0 & 0 & 0 \\
[1,2] & 0 & d & 0 & 0 & 0 & 0 & 0 \\
[1,3] & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
[1,4] & 0 & 0 & 0 & 0 & 0 & 0 & -d \\
[1,5] & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\
[1,6] & 0 & 0 & 0 & 0 & 0 & 0 & -2 \\
[1,7] & 0 & 0 & 0 & 0 & 0 & 0 & d \\
[2,1] & 0 & d & 0 & 0 & 0 & 0 & 0 \\
[2,2] & 0 & d\,\delta^2-d & 0 & 0 & 0 & 0 & 0 \\
[2,3] & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
[2,4] & 0 & d\,\delta & 0 & 0 & 0 & 0 & 0 \\
[2,5] & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\
[2,6] & 0 & 2\,\delta & 0 & 0 & 0 & 0 & 0 \\
[2,7] & 0 & -d\,\delta & 0 & 0 & 0 & 0 & 0 \\
[3,1] & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
[3,2] & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
[3,3] & 0 & 0 & 1+\delta^2 & 0 & 0 & 0 & 0 \\
[3,4] & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
[3,5] & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\
[3,6] & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
[3,7] & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
[4,1] & 0 & 0 & 0 & 0 & 0 & 0 & -d \\
[4,2] & 0 & d\,\delta & 0 & 0 & 0 & 0 & 0 \\
[4,3] & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
[4,4] & d & d & 0 & 0 & 0 & 0 & 0
\end{bmatrix}
\quad
\begin{bmatrix}
[4,5] & 0 & 0 & 0 & 0 & 0 & 0 & -1 \\
[4,6] & 2 & 2 & 0 & 0 & 0 & 0 & 0 \\
[4,7] & -d & -d & 0 & 0 & 0 & 0 & 0 \\
[5,1] & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\
[5,2] & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\
[5,3] & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\
[5,4] & 0 & 0 & 0 & 0 & 0 & 0 & -1 \\
[5,5] & 0 & 0 & 0 & 0 & 1 & 0 & 0 \\
[5,6] & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\
[5,7] & 0 & 0 & 0 & 0 & 0 & 0 & 1 \\
[6,1] & 0 & 0 & 0 & 0 & 0 & 0 & -2 \\
[6,2] & 0 & 2\,\delta & 0 & 0 & 0 & 0 & 0 \\
[6,3] & 0 & 0 & 2\,\delta & 0 & -2\,\delta-2\,\delta^3 & 1+\delta^2 & 0 \\
[6,4] & 2 & 2 & 0 & 0 & 0 & 0 & 0 \\
[6,5] & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\
[6,6] & 0 & 0 & 0 & 0 & 0 & 2\,\delta & 0 \\
[6,7] & -2 & -2 & 0 & 0 & 0 & 0 & 0 \\
[7,1] & 0 & 0 & 0 & 0 & 0 & 0 & d \\
[7,2] & 0 & -d\,\delta & 0 & 0 & 0 & 0 & 0 \\
[7,3] & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
[7,4] & -d & -d & 0 & 0 & 0 & 0 & 0 \\
[7,5] & 0 & 0 & 0 & 0 & 0 & 0 & 1 \\
[7,6] & -2 & -2 & 0 & 0 & 0 & 0 & 0 \\
[7,7] & d & d & 0 & 0 & 0 & 0 & 0
\end{bmatrix}$$

Hence, we obtain that the A-module E is generated by the A-endomorphisms f_i's defined by the 7 matrices P_i's of $Endo[1]$, i.e., $f_i(\pi(\lambda)) = \pi(\lambda\, P_i)$, where $\pi : D^{1\times 3} \longrightarrow M$ denotes the canonical projection onto M and λ is any element of $A^{1\times 3}$. Moreover, the generators f_i's of E satisfy the relations $Endo[2]\, F = 0$, where $F = (f_1 \ \dots \ f_7)^T$. Finally, the multiplication table T of the generators f_i's is the matrix $Endo[3]$ without the first column which corresponds to the indices (i, j) of the product $f_i \circ f_j$, namely, we have $F \otimes F = T\, F$, where $\otimes$ denotes the Kronecker product, namely, $F \otimes F = ((f_1 \circ F)^T \ \dots \ (f_7 \circ F)^T)^T$. Using $Endo[3]$, we can rewrite any polynomial in the f_i's with coefficients in A as an A-linear combination of the f_i's.

Let us try to find idempotent elements of E defined by idempotent matrices $P \in A^{3\times 3}$ and $Q \in A^{2\times 2}$, namely, $e \in E$ satisfying $e^2 = e$, where $e(\pi(\lambda)) = \pi(\lambda\, P)$, for all $\lambda \in A^{1\times 3}$, and $R\,P = Q\,R$, $P^2 = P$, $Q^2 = Q$:

```
> Idem:=IdempotentsMatConstCoeff(R,Endo[1],A,2);
```

$$Idem :=$$

$$[[\begin{bmatrix} -\delta^2 & 1/2\,\delta\,(1+\delta^2) & 0 \\ -2\,\delta & 1+\delta^2 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1+\delta^2 & -1/2\,\delta\,(1+\delta^2) & 0 \\ 2\,\delta & -\delta^2 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}],$$

$$[Ore\_algebra, [\text{“}diff\text{''}, dual\_shift], [t, s], [d, \delta], [t, s], [], 0, [], [], [t, s], [], [], [diff = [d, t], dual\_shift = [\delta, s]]]]$$

We obtain two non-trivial idempotent endomorphisms e_1 and e_2 of E respectively defined by the matrices $Idem[1,1]$ and $Idem[1,3]$. We note that we have $e_1 + e_2 = \mathrm{id}_M$. Let us consider e_1 defined by the following matrices $P = Idem[1,1]$ and $Q \in A^{2\times 2}$ satisfying $R\,P = Q\,R$:

```
> P:=Idem[1,1]; Q:=Factorize(Mult(R,P,A),R,A);
```

$$P := \begin{bmatrix} -\delta^2 & 1/2\,\delta\,(1+\delta^2) & 0 \\ -2\,\delta & 1+\delta^2 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad Q := \begin{bmatrix} 0 & 1/2\,\delta \\ 0 & 1 \end{bmatrix}$$

As we have $P^2 = P$ and $Q^2 = Q$, we know that the A-modules $\ker_A(.P)$, $\mathrm{im}_A(.P) = \ker_A(.(I_3 - P))$, $\ker_A(.Q)$ and $\mathrm{im}_A(.Q) = \ker_A(.(I_2 - Q))$ are projective, and thus, free by the Quillen-Suslin theorem. Let us compute bases of the corresponding modules. We try heuristic methods implemented in OreModules which do not require the use of the package QuillenSuslin:

```
> U1:=SyzygyModule(P,A): U2:=SyzygyModule(evalm(1-P),A):
> U:=stackmatrix(U1,U2);
> V1:=SyzygyModule(Q,A): V2:=SyzygyModule(evalm(1-Q),A):
> V:=stackmatrix(V1,V2);
```

$$U := \begin{bmatrix} -2 & \delta & 0 \\ 0 & 0 & 1 \\ -2\,\delta & 1+\delta^2 & 0 \end{bmatrix} \quad V := \begin{bmatrix} -2 & \delta \\ 0 & 1 \end{bmatrix}$$

We obtain that the two unimodular matrices U and V, i.e., $U \in \mathrm{GL}_3(A)$ and $V \in \mathrm{GL}_2(A)$, satisfy that the matrix $V\,R\,U^{-1}$ is block-diagonal:

```
> R_dec:=Mult(V,R,LeftInverse(U,A),A);
```

$$R\_dec := \begin{bmatrix} d - d\,\delta^2 & 2 & 0 \\ 0 & 0 & -d \end{bmatrix}$$

We can also use the command *HeuristicDecomposition* to directly obtain the previous result:

```
> HeuristicDecomposition(R,P,A)[1];
```

$$\begin{bmatrix} d - d\,\delta^2 & 2 & 0 \\ 0 & 0 & -d \end{bmatrix}$$

We can simplify R_dec by introducing the unimodular matrix X defined by:

```
> X:=diag(evalm([[0,1],[1/2,-(d-d*delta^2)/2]]),-1);
```

$$X := \begin{bmatrix} 0 & 1 & 0 \\ 1/2 & -1/2\,d + 1/2\,d\,\delta^2 & 0 \\ 0 & 0 & -1 \end{bmatrix}$$

Indeed, we have:

```
> Mult(R_dec,X,A);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 0 & d \end{bmatrix}$$

Therefore, if we consider the new matrix $W = X^{-1}\,U \in \mathrm{GL}_3(A)$ defined by

```
> W:=Mult(LeftInverse(X,A),U,A);
```

$$W := \begin{bmatrix} -2\,d + 2\,d\,\delta^2 & d\,\delta - d\,\delta^3 & 2 \\ -2 & \delta & 0 \\ 2\,\delta & -1-\delta^2 & 0 \end{bmatrix}$$

we then have the following simple decomposition of the matrix R:

```
> S:=Mult(V,R,LeftInverse(W,A),A);
```

$$S := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 0 & d \end{bmatrix}$$

Hence, we obtain that $M \cong A^{1\times 3}/(A^{1\times 2}\,S) \cong A \oplus A/(A\,d)$. Moreover, the linear system of differential time-delay equations $\ker_{\mathcal{F}}(R.)$, where $\mathcal{F}$ is an A-module (e.g., $\mathcal{F} = C^\infty(\mathbb{R})$) is equivalent to $\ker_{\mathcal{F}}(S.)$. In particular, an element $\zeta = (\zeta_1 \quad \zeta_2 \quad \zeta_3)^T \in \ker_{\mathcal{F}}(S.)$ satisfies $\zeta_1 = 0$, ζ_2 is arbitrary function of $\mathcal{F}$ and $\zeta_3 = c$ an arbitrary constant. Then, $\eta = W^{-1}\,\zeta$ is the general solution of the linear system $\ker_{\mathcal{F}}(R.)$.

We point out that the previous simple equivalent matrix S cannot be obtained by just noticing that the first row of R contains the invertible element -1 and post-multiplying R by the following elementary matrix Y

```
> Y:=matrix(3,3,[1,0,0,0,1,0,d,d*delta,-1]);
```

$$Y := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ d & -d\,\delta & -1 \end{bmatrix}$$

as we then obtain:

```
> L:=Mult(R,Y,A);
```

$$L := \begin{bmatrix} 0 & 0 & 1 \\ 2\,d\,\delta & -d\,\delta^2 - d & 0 \end{bmatrix}$$

We refer the reader to A. Fabiańska, A. Quadrat, "Applications of the Quillen-Suslin theorem in multidimensional systems theory", chapter of the book *Gröbner Bases in Control Theory and Signal Processing*, H. Park and G. Regensburger (Eds.), Radon Series on Computation and Applied Mathematics 3, de Gruyter publisher, 2007, 23-106, for different algorithms which simplify the presentation matrices. Indeed, the previous computation only shows that we have:

$$M \cong A^{1\times 3}/(A^{1\times 2}\,L) \cong A^{1\times 2}/(A\,(2\,d\,\delta \quad -d\,(\delta^2+1)).$$

Using the equivalent presentation matrix L of M, we then need to compute $t(M)$ and $M/t(M)$ as explained in F. Chyzak, A. Quadrat, D. Robertz, "Effective algorithms for parametrizing linear control systems over Ore algebras", *Appl. Algebra Engrg. Comm. Comput.*, 16 (2005), 319-376, to get that $t(M) \cong A/(A\,d)$ and $M/t(M) \cong A$ and to combine these results with the particular fact that $M \cong t(M) \oplus M/t(M)$ to find again that $M \cong A \oplus A/(A\,d)$. However, all these information are obtained in one step using the previous decomposition approach.

Let us study the A-module structure $A^{1\times 7}/(A^{1\times 6}\,Endo[2])$ of the endomorphism ring E.

```
> ext1:=Exti(Involution(Endo[2],A),A,1);
```

$$
ext1 := \left[\begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & d & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & d & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 1 & 0 & 0 & 0 & 0 & \delta \\ \delta & 0 & 0 & 0 & 0 & 0 & 1 \\ -1 & 0 & 0 & 0 & d & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 2\,\delta & -1 & 0 \\ 0 & 0 & 0 & 0 & 0 & d & 2 \end{bmatrix}, \begin{bmatrix} -d \\ d - d\,\delta^2 \\ 0 \\ -d\,\delta \\ -1 \\ -2\,\delta \\ d\,\delta \end{bmatrix} \right]
$$

We obtain that the endomorphisms $t_1 = f_3$ and $t_2 = 2\,\delta\, f_5 - f_6$ generate the torsion A-module $t(E)$. We note that $f_5 = \mathrm{id}_M$, a fact showing that $t_2 = 2\,\delta\,\mathrm{id}_M - f_6$. In particular, we obtain that every element in $t_1(M)$ or in $t_2(M)$ define a torsion element of M. Moreover, the A-module $E/t(E)$ is finitely presented by the second matrix $ext1[2]$ of $ext1$, i.e., $E/t(E) = A^{1\times 7}/(A^{1\times 7}\, ext1[2])$. We also have $E/t(E) \cong A^{1\times 7}\, ext1[3]$.

```
> T:=LeftInverse(ext1[3],A);
```

$$
T := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & -1 & 0 \end{bmatrix}
$$

As the matrix $ext1[3]$ admits a left-inverse of A, we obtain that $E/t(E) \cong A$, i.e., $E/t(E)$ is a free A-module of rank 1. In particular, the short exact sequence of A-modules

$$
0 \longrightarrow t(E) \xrightarrow{\iota} E \xrightarrow{\rho} E/t(E) \longrightarrow 0 \tag{1}
$$

splits and we obtain $E \cong t(E) \oplus E/t(E) \cong t(E) \oplus A$. Let us now study the A-module $t(E)$:

```
> K:=stackmatrix(Factorize(Endo[2],ext1[2],A),SyzygyModule(ext1[2],A));
```

$$
K := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & d & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \\ 0 & -2 & -2\,\delta & 0 & 0 & d & 1 \end{bmatrix}
$$

We obtain that $t(E) \cong A^{1\times 7}/(A^{1\times 7}\, K)$. Using the special structure of the matrix K, we get that $t(E) \cong A/(A\, d) \oplus A/(A\, d)$, which shows that:

$$
E \cong [A/(A\,d)]^2 \oplus A. \tag{2}
$$

(2) is consistent with the fact that $M \cong A^{1\times 3}/(A^{1\times 2}\, S) \cong A \oplus A/(A\,d)$ which implies that:

$$
\begin{aligned}
E &= \mathrm{end}_A(M) \cong \hom_A(A \oplus A/(A\,d), A \oplus A/(A\,d)) \\
&\cong \mathrm{end}_A(A) \oplus \hom_A(A, A/(A\,d)) \oplus \hom_A(A/(A\,d), A) \oplus \mathrm{end}_A(A/(A\,d)).
\end{aligned}
$$

We have $\mathrm{end}_A(A) \cong A$, $\hom_A(A, A/(A\,d)) \cong A/(A\,d)$ and $\hom_A(A/(A\,d), A) = 0$ because $A/(A\,d)$ is a torsion A-module and A is torsion-free. Moreover, we have $\mathrm{end}_A(A/(A\,d)) \cong A/(A\,d)$, which proves (2).

Using the following notations $F = (f_1\ \dots\ f_7)^T$ and $G = (g_1\ \dots\ g_7)^T$

```
> F:=evalm([seq([f[i]],i=1..nops(Endo[1]))]);
> G:=evalm([seq([g[i]],i=1..rowdim(ext1[2]))]);
```

$$F := \begin{bmatrix} f_1 \\ f_2 \\ f_3 \\ f_4 \\ f_5 \\ f_6 \\ f_7 \end{bmatrix} \quad G := \begin{bmatrix} g_1 \\ g_2 \\ g_3 \\ g_4 \\ g_5 \\ g_6 \\ g_7 \end{bmatrix}$$

the generators f_i's of the A-module E satisfy the relations $Endo[2]\, F = 0$, namely,

```
> evalm(Endo[2]&*F)=evalm([[0]$rowdim(Endo[2])]);
```

$$\begin{bmatrix} f_1 + f_2 + \delta\, f_7 \\ \delta\, f_1 + f_7 \\ -f_1 + d\, f_5 \\ d\, f_3 \\ f_4 + f_7 \\ d\, f_6 + 2\, f_7 \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

and the generators g_i's of $E/t(E)$ satisfy the equation $ext1[2]\, G = 0$:

```
> evalm(ext1[2]&*G)=evalm([[0]$rowdim(ext1[2])]);
```

$$\begin{bmatrix} g_1 + g_2 + \delta\, g_7 \\ \delta\, g_1 + g_7 \\ -g_1 + d\, g_5 \\ g_3 \\ g_4 + g_7 \\ 2\,\delta\, g_5 - g_6 \\ d\, g_6 + 2\, g_7 \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

Using the split exact sequence of A-modules $A^{1\times 7} \xrightarrow{.ext1[2]} A^{1\times 7} \xrightarrow{.ext1[3]} A \longrightarrow 0$, we obtain the following injective parametrization of the generators g_i's of $E/t(E)$

```
> evalm(G)=evalm(ext1[3]*h);
```

$$\begin{bmatrix} g_1 \\ g_2 \\ g_3 \\ g_4 \\ g_5 \\ g_6 \\ g_7 \end{bmatrix} = \begin{bmatrix} -d\, h \\ (d - d\,\delta^2)\, h \\ 0 \\ -d\,\delta\, h \\ -h \\ -2\,\delta\, h \\ d\,\delta\, h \end{bmatrix}$$

where h is defined by

```
> h=evalm(T&*G)[1,1];
```

$$h = -g_5$$

i.e., we have $E/t(E) \cong A\, g_5$ and $\mathrm{ann}_A(g_5) = 0$. Moreover, we have $t(E) \cong A\, t_1 \oplus A\, t_2$, where $\mathrm{ann}_A(t_1) = \mathrm{ann}_A(t_2) = A\, d$, which shows that $E \cong A\, g_5 \oplus A\, t_1 \oplus A\, t_2$.

To finish, we can explicitly describe the previous isomorphism. In order to do that, we first compute a generalized inverse Z of $ext1[2]$ over A:

```
> Z:=GeneralizedInverse(ext1[2],A);
```

$$Z := \begin{bmatrix} 0 & 0 & -1 & 0 & 0 & 0 & 0 \\ 1 & -\delta & 1-\delta^2 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & -1 & -\delta & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -1 & 0 \\ 0 & 1 & \delta & 0 & 0 & 0 & 0 \end{bmatrix}$$

We can check that we have $ext1[2]\, Z\, ext1[2] = ext1[2]$. Let us denote by $H = I_7 - Z\, ext1[2]$:

```
> H:=evalm(1-Mult(Z,ext1[2],A));
```

$$H := \begin{bmatrix} 0 & 0 & 0 & 0 & d & 0 & 0 \\ 0 & 0 & 0 & 0 & d\,(\delta^2-1) & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & d\,\delta & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 2\,\delta & 0 & 0 \\ 0 & 0 & 0 & 0 & -d\,\delta & 0 & 0 \end{bmatrix}$$

Using the fact that $ext1[2]\, H = 0$, we obtain that the A-morphism $\sigma : E/t(E) \longrightarrow E$ defined by $\sigma(\pi'(\lambda)) = \pi(\lambda\, H)$, where $\pi' : A^{1\times 7} \longrightarrow E/t(E)$ denotes the projection onto $E/t(E)$ and λ is an element of $A^{1\times 7}$, satisfies $\rho \circ \sigma = \mathrm{id}_{E/t(E)}$. For more details, see Theorem 4 of A. Quadrat, D. Robertz, "Parametrizing all solutions of uncontrollable multidimensional linear systems", *Proceedings of* 16^{th} *IFAC World Congress*, Prague (Czech Republic), 04-08/07/05. We find again that the short exact sequence (1) splits. The A-morphism σ is defined by:

$$\begin{cases} \sigma(g_1) = d\, f_5, \\ \sigma(g_2) = d\,(\delta^2-1)\, f_5, \\ \sigma(g_3) = 0, \\ \sigma(g_4) = d\,\delta\, f_5, \\ \sigma(g_5) = f_5, \\ \sigma(g_6) = 2\,\delta\, f_5, \\ \sigma(g_7) = -d\,\delta\, f_5. \end{cases}$$

Using the relations between the generators f_i's of the A-module E, we obtain that the A-morphism $\chi : \mathrm{id}_E - \sigma \circ \rho : E \longrightarrow E$ is defined by:

$$\begin{cases} \chi(f_1) = f_1 - d\, f_5 = 0, \\ \chi(f_2) = f_2 - d\,(\delta^2-1)\, f_5 = 0, \\ \chi(f_3) = f_3 = t_1, \\ \chi(f_4) = f_4 - d\,\delta\, f_5 = 0, \\ \chi(f_5) = f_5 - f_5 = 0, \\ \chi(f_6) = f_6 - 2\,\delta\, f_5 = -t_2, \\ \chi(f_7) = f_7 - d\,\delta\, f_5 = 0. \end{cases}$$

Hence, if we define the A-morphism $\kappa : E \longrightarrow t(E)$ by

$$\begin{cases} \kappa(f_1) = 0, \\ \kappa(f_2) = 0, \\ \kappa(f_3) = t_1, \\ \kappa(f_4) = 0, \\ \kappa(f_5) = 0, \\ \kappa(f_6) = -t_2, \\ \kappa(f_7) = 0, \end{cases}$$

we then get that $\mathrm{id}_E = \sigma \circ \rho + \iota \circ \kappa$. Therefore, using the fact that $f_5 = \mathrm{id}_M$, we obtain

$$\begin{cases} f_1 = d\,\mathrm{id}_M, \\ f_2 = d\,(\delta^2 - 1)\,\mathrm{id}_M, \\ f_3 = t_1, \\ f_4 = d\,\delta\,\mathrm{id}_M, \\ f_5 = \mathrm{id}_M, \\ f_6 = 2\,\delta\,\mathrm{id}_M - t_2, \\ f_7 = -d\,\delta\,\mathrm{id}_M, \end{cases} \tag{3}$$

a fact showing that the generators f_i's of E can be expressed in terms of id_M and $t_1 = f_3$ and $t_2 = 2\,\delta\,\mathrm{id}_M - f_6$ and $\{\mathrm{id}_M, t_1, t_2\}$ generates the A-module E. In particular, using the multiplication table *Endo*[3] and (3), we can easily obtain the following small multiplication table for the new family of generators $\{\mathrm{id}_M, t_1, t_2\}$ (compare with *Endo*[3]):

$$\begin{cases} t_1 \circ t_1 = (1 + \delta^2)\,t_1 \\ t_1 \circ t_2 = 2\,\delta\,t_1, \\ t_2 \circ t_1 = (1 + \delta^2)\,t_2, \\ t_2 \circ t_2 = 4\,\delta\,t_2 - 2\,\delta\,(2\,\delta - 1)\,\mathrm{id}_M, \\ t_i \circ \mathrm{id}_M = \mathrm{id}_M \circ t_i = t_i,\ i = 1, 2, \end{cases}$$

Using it, we can rewrite any polynomial in the f_i's with coefficients in A in terms of an A-linear combination of id_M, t_1 and t_2.